


Simulationsbeispiel 2.1: R - $2R$ -Netzwerk



Bibliotheken aus [2a]
 Gleichstrom-Analyse
 Arbeitspunkt-Analyse
 (Bias Point)

Bild 2.142: Schaltung zur Simulation der prinzipiellen Wirkungsweise eines R - $2R$ -Netzwerkes

* Schematics Netlist SB_2.1 *

```
V_Uq      1  0  32V
R_R1      1  2  1k
R_R2      2  0  2k
R_R3      2  3  1k
R_R4      3  0  2k
R_R5      3  4  1k
R_R6      4  0  2k
R_R7      4  5  1k
R_R8      5  0  2k
R_Ra      5  0  2k
```

** Analysis setup **

```
.TEMP      20
.OP
```

*** RESUMING SB_2_1.cir ***


```
.probe
.END
```

- Aufruf der **Lösungen** (nur bei einer gezeichneten Schaltung) über:

> Enable Bias Voltage Display < [Werte der Knotenpotentiale]
 > Enable Bias Current Display < [Werte der Zweigströme]

Weitere Informationen (ohne Schaltung): *Analysis* → *Examine Output* (Output-File)

Simulationsbeispiel 2.2: HF-Ersatzschaltbild des ohmschen Widerstandes



Bibliotheken aus [2a]

AC-Analyse
 AC- Sweep (Decade)

Bild 2.143: Simulation der HF-Ersatzschaltung eines ohmschen Widerstandes

* Schematics Netlist SB_2.2 *

```
V_Uq      1  0  AC  10V SIN 0 10V 1kHz 0 0 0
R_R       A  2  100
L_Lr      2  0  185uH
C_Cp      A  0  10pF
R_RH      1  A  1m
```

** Analysis setup **

```
.AC      DEC      10000      100 100MEG
.TEMP    20
.OP
```

*** RESUMING SB_2_2.cir ***

```
.probe
.END
```


- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → *Add Trace* V(1)/I(R_RH) [Frequenzgang des Widerstandes R_{AB}]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)



Simulationsbeispiel 2.3: Transformator



Bibliotheken aus [2a]

DC-Analyse
 DC-Main-Sweep

Bild 2.146: Simulationsschaltung zur Aufnahme der Magnetisierungskennlinie

* Schematics Netlist SB_2.3 *

```
I_Iq      0  P  DC  1A
K_TX1     L1_TX1 L2_TX1 1  K528T500_3C8
L1_TX1    P    0    10
L2_TX1    S    0    0.1
R_Ra      S    0    10MEG
```

** Analysis setup **

```
.DC      LIN      I_Iq      0  10  1m
.TEMP    20
.OP
```

*** RESUMING SB_2_3.cir ***

```
.probe
.END
```


- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → *Add Trace* B(K_TX1) [magnetische Flussdichte]

Anmerkung: Die Achsen müssen noch nachbearbeitet werden (Einheiten nach DIN).

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

Simulationsbeispiel 2.4: Kenngrößen einer Diode



Bibliotheken aus [2a]

DC-Analyse
 DC-Main-Sweep

Bild 2.152: Einstellung und Kenngrößen eines Arbeitspunktes der Diode 1N 4002

Anmerkung: Mit der Netzliste wird nur die Strom-Spannungs-Kennlinie der Diode simuliert (also: $R_i = 0$).

* Schematics Netlist SB_2.4 *

```
V_Uq      1  0  1V
D_V1      1  0  D1N4002
```

** Analysis setup **

```
.DC      LIN V_Uq  -101  2  1m
.TEMP    20
.OP
```

*** RESUMING SB_2_4.cir ***


```
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → *Add Trace* I(V1) [Strom-Spannungs-Kennlinie der Diode]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

Simulationsbeispiel 2.5: Gleichrichtung



Bibliotheken aus [2a]

Transienten-Analyse

Bild 2.155: Schaltung zur Simulation eines Brückengleichrichters

* Schematics Netlist SB_2.5 *

```
V_UP      E  G  PULSE -5V 5V 0 1ms 0.1ps 0.1ps 1ms
D_V4      0  E  D1N4148
D_V1      E  F  D1N4148
D_V2      G  F  D1N4148
D_V3      0  G  D1N4148
R_RM      F  0  1k
```

** Analysis setup **

```
.TRAN      0 5ms 0 1u
.TEMP      20
.OP
```

*** RESUMING SB_2_5.cir ***


```
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → *Add Trace* V(F) [Zeitfunktion der Spannung u_{FH}]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

Simulationsbeispiel 2.6.a: Extremwertgatter (Höchstwert)



Bibliotheken aus [2a]

DC-Analyse
 DC-Main-Sweep

Bild 2.157: Höchstwertgatter mit Eingangs- und Ausgangssignalen

* Schematics Netlist SB_2.6.a *

```
V_Uq1      x1  0  5V
V_Uq2      x2  0  1V
X_VD3      x1  x2  y  Dbreak3
R_Ra       y   0  1k
```

** Analysis setup **

```
.DC          LIN V_Uq2  0  10  1m
.TEMP       20
.OP
```

*** RESUMING SB_2.6_a.cir ***


```
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → Add Trace V(x1) V(x2) V(y) [Verläufe der Spannungen]

Weitere Informationen: Analysis → Examine Output (Output-File)

Simulationsbeispiel 2.6.b: Extremwertgatter (Tiefstwert)



Bibliotheken aus [2a]

DC-Analyse
 DC-Main-Sweep

Bild 2.158: Simulation eines Tiefstwertgatters

* Schematics Netlist SB_2.6.b *

```
V_UB      1  0  10V
V_Uq1     x1 0   5V
V_Uq2     x2 0   1V
D_V1      y  x1  D1N4148
D_V2      y  x2  D1N4148
R_R1      1  y   1k
R_Ra      y  0  10k
```

** Analysis setup **

```
.DC          LIN V_Uq2  0  10  1m
.TEMP       20
.OP
```

*** RESUMING SB_2_6_b.cir ***


```
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → *Add Trace* V(x1) V(x2) V(y) [Verläufe der Spannungen]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

Simulationsbeispiel 2.7.a: Spannungsstabilisierung (Bestimmung R_V)



Bibliotheken aus [2a]

DC-Analyse
 DC-Main-Sweep

Bild 2.160.a: Variation des Vorwiderstandes in einer Stabilisierungsschaltung mit Z-Diode

* Schematics Netlist SB_2.7.a *

```
V_Uq      1  0  10V
D_V1      0  2  D1N750
R_R1      1  2  {RV}
R_Ra      2  0  500
.PARAM    RV=100
```

** Analysis setup **

```
.DC        LIN PARAM    RV 10 1k 1
.TEMP      20
.OP
```

*** RESUMING SB_2_7_a.cir ***


```
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → *Add Trace* V(2) [Spannungsverlauf bei Variation des Vorwiderstandes]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

Simulationsbeispiel 2.7.b: Spannungsstabilisierung (Arbeitspunkt)



Bibliotheken aus [2a]

DC-Analyse
 Arbeitspunkt-Analyse
 (Bias Point)

Bild 2.160.b: Arbeitspunkt in einer Stabilisierungsschaltung mit Z-Diode

In diesem Beispiel gilt: $U_q = 10 \text{ V}$ und $R_a = 500 \text{ } \Omega$ sowie $R_1 = R_V = 89 \text{ } \Omega = \text{const.}$

* Schematics Netlist SB_2.7.b *

```
V_Uq      1  0  10V
D_V1      0  2  D1N750
R_R1      1  2  89
R_Ra      2  0  500
```

** Analysis setup **

```
.TEMP      20
.OP
```

*** RESUMING SB_2_7_b.cir ***

```
.probe
.END
```


• Aufruf der **Lösungen** (nur bei einer gezeichneten Schaltung) über:

```
> Enable Bias Voltage Display <      [Werte der Knotenpotentiale]
> Enable Bias Current Display <     [Werte der Zweigströme]
```

Weitere Informationen (ohne Schaltung): *Analysis* → *Examine Output* (Output-File)



Simulationsbeispiel 2.8: Kapazitätsdiode



Bibliotheken aus [2a]

AC-Analyse
 AC-Main-Sweep (Decade)

Bild 2.166: Schaltungsentwurf zur Gestaltung einer Netzliste

* Schematics Netlist SB_2.8 *

```
V1 1 0 AC 10V SIN 0 10V 1kHz 0 0 0
R_R1 2 0 100k
R_R2 3 0 100k
R_R3 4 0 100k
R_R4 5 0 100k
D1 1 2 DIODE1
D2 1 3 DIODE2
D3 1 4 DIODE3
D4 1 5 DIODE4
.MODEL DIODE1 D (IS = 1.365p RS = 1 CJO = 14.93p)
.MODEL DIODE2 D (IS = 1.365p RS = 1 CJO = 10p)
.MODEL DIODE3 D (IS = 1.365p RS = 1 CJO = 5p)
.MODEL DIODE4 D (IS = 1.365p RS = 1 CJO = 1p)
```

** Analysis setup **

```
.AC DEC 1000 1K 10MEG
.TEMP 20
.OP
```

*** RESUMING SB_2_8.cir ***


```
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → *Add Trace* I(D1) I(D2) I(D3) I(D4) [Frequenzgänge des Ströme]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

Simulationsbeispiel 2.9: Schottky-Diode



Bibliotheken aus [2a]

DC-Analyse
 DC-Main-Sweep

Bild 2.168: Nachbildung der Strom-Spannungs-Kennlinie der SCHOTTKY-Diode

* Schematics Netlist SB_2.9 *

```
V_Uq      1  0  1V
V_US1     0  2  280mV
D_D1      1  2  Dbreak
D_V1      1  0  MBD101
.MODEL Dbreak  D
+ (IS=1e-14  CJO=.1pF  RS=.1  BV=4.7  IBV=10u)

** Analysis setup **
.DC       LIN V_Uq  -6  1  1m
.TEMP     20
.OP

*** RESUMING SB_2_9.cir ***
.probe
.END
```


- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → *Add Trace* I(D1) I(V1) [Strom-Spannungs-Kennlinien]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

Simulationsbeispiel 2.10.a: Parasitäre Effekte im HF-Bereich

(NEU: 2020)



Bibliotheken aus [5]: model .subckt RHF-X

AC-Analyse
 AC- Sweep (Decade)
 Parametric-Sweep

Bild 2.171.a: Simulationsschaltungen mit Angabe der Werte der parasitären Elemente (hier: ohmscher Widerstand)

* Schematics Netlist SB_2.10.a *

```
.subckt RHF-X 1 2 params: r=50 cr=0.2p lr=3n
r1 1 3 {r}
l1 2 3 {lr}
c1 1 2 {cr}
.ends

V_Uq A 0 DC 0 AC 1V
+ SIN 0V 1V 100kHz 0 0 0
R_RH A 1 1u
X_RHF 1 0 RHF-X params: r={n*50} lr=3nH cr=0.2pF
.PARAM n=1

** Analysis setup **
.AC DEC 10000 100MEG 30G
.STEP PARAMn LIST 1 2
.TEMP 20
.OP

*** RESUMING SB_2_10_a.cir ***
.probe
.END
```


• Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

- | | | |
|----------------------|---------------|------------------------------|
| a) Trace → Add Trace | I(RH) | [Frequenzgänge des Stromes] |
| b) Trace → Add Trace | V(1)/I(RH) | [Frequenzgänge der Impedanz] |
| c) Trace → Add Trace | P(V(1)/I(RH)) | [Phasenfrequenzgänge] |

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

Simulationsbeispiel 2.10.b: Parasitäre Effekte im HF-Bereich

(NEU: 2020)



Bibliotheken aus [5]: model .subckt CHF-X

AC-Analyse
 AC- Sweep (Decade)
 Parametric-Sweep

Bild 2.171.b: Simulationsschaltungen mit Angabe der Werte der parasitären Elemente (hier: Kondensator)

* Schematics Netlist SB_2.10.b *

```
.subckt CHF-X 1 2 params: c=1n lc=3n rc=0.2
c1 1 3 {c}
l1 3 4 {lc}
r1 4 2 {rc}
rd1 1 2 1e12
.ends

V_Uq      A      0      DC      0      AC      1V
+ SIN 0V 1V 100kHz 0 0 0
R_RH      A      1      1u
X_CHF     1      0      CHF-X params: c={n*100pF} lc=3nH rc=0.2
.PARAM    n=1

** Analysis setup **
.AC DEC 10000 1MEG      100G
.STEP PARAM    n      LIST 0.1 1
.TEMP      20
.OP

*** RESUMING SB_2_10_b.cir ***
.probe
.END
```


• Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

- a) *Trace* → *Add Trace* I(RH) [Frequenzgänge des Stromes]
- b) *Trace* → *Add Trace* V(1)/I(RH) [Frequenzgänge der Impedanz]
- c) *Trace* → *Add Trace* P(V(1)/I(RH)) [Phasenfrequenzgänge]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

Simulationsbeispiel 2.10.c: Parasitäre Effekte im HF-Bereich

(NEU: 2020)



Bibliotheken aus [5]: model .subckt LHF-X

AC-Analyse
 AC- Sweep (Decade)
 Parametric-Sweep

Bild 2.171.c: Simulationsschaltungen mit Angabe der Werte der parasitären Elemente (hier: Spule)

* Schematics Netlist SB_2.10.c *

```
.subckt LHF-X 1 2 params: l=100n cl=0.2p krl=1.2e-4
l1 1 3 {l}
g1 2 3 laplace { v(2,3) } = { 1/(krl*sqrt(1+0.31831*s)) }
c1 1 2 {cl}
rd1 1 2 1e12
.ends

V_Uq      A      0      DC      0      AC      1V
+ SIN 0V 1V 100kHz 0 0 0
R_RH      A      1      1u
X_LHF     1      0      LHF-X params: l={n*100nH} cl=0.2pF krl={1200*{n*100nH}}
.PARAM    n=1

** Analysis setup **
.AC DEC 10000 10k 30G
.STEP PARAM n LIST 1 10
.TEMP 20
.OP

*** RESUMING SB_2_10_c.cir ***
.probe
.END
```

• Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

- | | | |
|----------------------|---------------|------------------------------|
| a) Trace → Add Trace | I(RH) | [Frequenzgänge des Stromes] |
| b) Trace → Add Trace | V(1)/I(RH) | [Frequenzgänge der Impedanz] |
| c) Trace → Add Trace | P(V(1)/I(RH)) | [Phasenfrequenzgänge] |

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)